Response of Single Degree-of-Freedom System—RESP

The program RESP (**Resp**onse of Single Degree-of-Freedom System) is a subroutine subprogram that calculates the absolute acceleration response, relative velocity response, and relative displacement response time histories of a single mass damped system by integrating the equations of motion of the system when given the time history of the ground motion acceleration, and also calculates the maximum values of these responses.

RESP （**Resp**onse of Single Degree-of-Freedom System）

【Purpose】

To calculate the absolute acceleration response, relative velocity response, and relative displacement response time histories and their maximum values for a given ground acceleration time history for a single degree of freedom system with a given natural circular frequency and damping factor.

【Usage】

(1) How to connect

```
CALL RESP (H, W, DT, NN, DDY, ACC, VEL, DIS, ND, SA, SV, SD)
```

Argument	Type	Parameter in calling program	Return Parameter
H	R	Damping factor (no dimension)	Unchanged
W	R	Natural circular frequency (unit : rad/sec)	Unchanged
DT	R	Time increment (unit : sec)	Unchanged
NN	I	Number of data	Unchanged
DDY	R 1-D array (ND)	Time history of ground acceleration (unit : Gal)	Unchanged
ACC	R 1-D array (ND)	No need to input here	Absolute acceleration response time history (unit : Gal=cm/sec^2)
VEL	R 1-D array (ND)	No need to input here	Relative velocity response time history (unit : cm/sec)
DIS	R 1-D array (ND)	No need to input here	Relative displacement response time history (unit : cm)

ND	I	Dimension size of DDY, ACC, VEL and DIS in calling program	Unchanged
SA	R	No need to input here	Maximum value of absolute acceleration response (unit : Gal)
SV	R	No need to input here	Maximum value of relative velocity response (unit : cm/sec)
SD	R	No need to input here	Maximum value of relative displacement response (unit : cm)

(2) Necessary subroutines and function subprograms

None

【Program List】

```
C  * * * * * * * * * * * * * * * * * * * * * * * * * *                 RESP   1
C     SUBROUTINE FOR RESPONSE OF SINGLE-DOF SYSTEM                      RESP   2
C  * * * * * * * * * * * * * * * * * * * * * * * * * *                 RESP   3
C                                                                       RESP   4
C                                CODED BY Y.OHSAKI                      RESP   5
C                                                                       RESP   6
C     PURPOSE                                                           RESP   7
C       TO COMPUTE ABSOLUTE ACCELERATION, RELATIVE VELOCITY AND RELA-   RESP   8
C       TIVE DISPLACEMENT RESPONSE TIME HISTORIES AND THEIR MAXIMA OF   RESP   9
C       A SINGLE DEGREE-OF-FREEDOM SYSTEM WITH GIVEN NATURAL FREQUENCY  RESP  10
C       AND DAMPING FACTOR EXCITED BY GIVEN ACCELERATIONS               RESP  11
C                                                                       RESP  12
C     USAGE                                                             RESP  13
C       CALL RESP(H,W,DT,NN,DDY,ACC,VEL,DIS,ND,SA,SV,SD)                RESP  14
C                                                                       RESP  15
C     DESCRIPTION OF ARGUMENTS                                          RESP  16
C       H       - DAMPING FACTOR IN DECIMAL FRACTION                    RESP  17
C       W       - NATURAL CIRCULAR FREQUENCY IN RAD/SEC                 RESP  18
C       DT      - TIME INCREMENT IN TIME HISTORIES IN SEC               RESP  19
C       NN      - TOTAL NUMBER OF DATA IN TIME HISTORIES                RESP  20
C       DDY(ND) - GIVEN ACCELERATION TIME HISTORY IN GALS               RESP  21
C       ACC(ND) - ABSOLUTE ACCELERATION RESPONSE TIME HISTORY IN GALS   RESP  22
C       VEL(ND) - RELATIVE VELOCITY RESPONSE TIME HISTORY IN KINES      RESP  23
C       DIS(ND) - RELATIVE DISPLACEMENT RESPONSE TIME HISTORY IN CENTI- RESP  24
C                 METERS                                                RESP  25
C       ND      - DIMENSION OF DDY,ACC,VEL,DIS IN CALLING PROGRAM       RESP  26
C       SA      - MAX. ABSOLUTE ACCELERATION RESPONSE IN GALS           RESP  27
C       SV      - MAX. RELATIVE VELOCITY RESPONSE IN KINES              RESP  28
C       SD      - MAX. RELATIVE DISPLACEMENT RESPONSE IN CENTIMETERS    RESP  29
C                                                                       RESP  30
C     SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED                     RESP  31
C       NONE                                                            RESP  32
C                                                                       RESP  33
      SUBROUTINE RESP(H,W,DT,NN,DDY,ACC,VEL,DIS,ND,SA,SV,SD)            RESP  34
C                                                                       RESP  35
```

```
      DIMENSION DDY(ND),ACC(ND),VEL(ND),DIS(ND)
C
      W2=W*W
      HW=H*W
      WD=W*SQRT(1.-H*H)
      WDT=WD*DT
      E=EXP(-HW*DT)
      CWDT=COS(WDT)
      SWDT=SIN(WDT)
      A11= E*(CWDT+HW*SWDT/WD)
      A12= E*SWDT/WD
      A21=-E*W2*SWDT/WD
      A22= E*(CWDT-HW*SWDT/WD)
      SS=-HW*SWDT-WD*CWDT
      CC=-HW*CWDT+WD*SWDT
      S1=(E*SS+WD)/W2
      C1=(E*CC+HW)/W2
      S2=(E*DT*SS+HW*S1+WD*C1)/W2
      C2=(E*DT*CC+HW*C1-WD*S1)/W2
      S3=DT*S1-S2
      C3=DT*C1-C2
      B11=-S2/WDT
      B12=-S3/WDT
      B21=(HW*S2-WD*C2)/WDT
      B22=(HW*S3-WD*C3)/WDT
      ACC(1)=2.*H*W*DDY(1)*DT
      VEL(1)=-DDY(1)*DT
      DIS(1)=0.
      DX=VEL(1)
      X=0.
      SA=0.
      SV=0.
      SD=0.
      DO 110 M=2,NN
      DXF=DX
      XF=X
      DDYM=DDY(M)
      DDYF=DDY(M-1)
      X= A12*DXF+A11*XF+B12*DDYM+B11*DDYF
      DX=A22*DXF+A21*XF+B22*DDYM+B21*DDYF
      DDX=-2.*HW*DX-W2*X
      ACC(M)=DDX
      VEL(M)=DX
      DIS(M)=X
      SA=AMAX1(SA,ABS(DDX))
      SV=AMAX1(SV,ABS(DX))
      SD=AMAX1(SD,ABS(X))
  110 CONTINUE
      RETURN
      END
```

【Example】

Compute the absolute acceleration response, relative velocity response, and relative displacement response for a single mass damping system with an undamped eigen period of 0.3 sec and a damping factor of h=5% subjected to the acceleration time history of the El Centro seismic wave (EQ.01) as ground motion. The results, which are stored in the arrays *ACC*, *VEL*, and *DIS*, respectively, are shown in the figures below.

```
C
      DIMENSION DDY(800),ACC(800),VEL(800),DIS(800)
      DATA H/0.05/,T/0.3/
C
      READ(5,501) DT,NN,(DDY(M),M=1,NN)
C
      W=6.283185/T
      CALL RESP (H,W,DT,NN,DDY,ACC,VEL,DIS,800,SA,SV,SD)
      STOP
  501 FORMAT(T51,F10.0,I10/(8F10.0))
      END
```

Output：